


PSPICE example

Investigation of operational
amplifier parameters

Gdańsk, October 2019

Amplifier schematic



Parameters to be found (in open loop without load)

- Input referred offset voltage (V_{IOFF}),
- Voltage gain (differential) (A_D),
- Common mode gain (A_{CM}),
- Common Mode Rejection Ratio ($CMRR$),
- Power Supply Rejection Ratio ($PSRR$)
- Input resistance (R_{ID}),
- Output resistance (R_{OUT}),
- Passband, GBW, phase and gain margin
- Slew Rate factor (SR)

Circuit netlist / SPICE commands – plain version with testbench inside one file

Operational Amplifier example

*** Op Amp circuitry *****

Q1 1 in_m 12 qnpn

Q2 2 in_p 12 qnpn

Q3 12 5 3 qnpn

Q4 out 5 4 qnpn 10

Q5 5 5 5b qnpn 10

Q6 1 1 6 qnpn

Q7 2 1 7 qnpn

Q8 out 2 8 qnpn 20

R3 3 Vee 600

R4 4 Vee 60

R5a 0 5 4.24k

R5b 5b Vee 60

R6 Vcc 6 1k

R7 Vcc 7 1k

R8 Vcc 8 50

Cc out 2 1pF

*** Power supply *****

Vcc Vcc 0 5V

Vee 0 Vee 5V

continued ...

*** Input source *****

Vid in_p in_m {vioff} ac 1

Ecm in_p cm in_p in_m 0.5

Vcm cm 0 0V

*** Load *****

Rload out 0 1G

*** Library files (models) ***

.lib modele8.lib

*** simulations commands and parameters ***

.dc vid -100m 100m .01m

.param vioff=8.278mV

.ac dec 100 1 1G

*** Options and output set *****

.probe

.end

Circuit netlist / SPICE commands – tested circuit and testbench splited into separate files

Testbench file:

```
Operational Amplifier example      continued ...

.lib op_amp_sub.lib                *** Load *****
X1 inp inm out vcc vee op_amp      Rload out 0 1G

*** Power supply *****          *****
Vcc Vcc 0 5V                       *** Library files (models) ***
Vee 0 Vee 5V                       .lib modele8.lib
*****                              *****

*** Input source *****          *** simulations and cmds ***
Vid inp inm {vioff} ac 1            .dc vid -100m 100m .01m
Ecm inp cm inp inm 0.5             .param vioff=8.278mV
Vcm cm 0 0V                        .ac dec 100 1 1G
*****                              *****

*** OUT resistance measurement ***  *** Options and output set **
Iro 0 out DC 0                     .probe
*****                              *****

                                     .end
```

Tested circuit file:

```
*** Operational Amplifier subcircuit
.subckt op_amp in_p in_m out vcc vee

*** Op Amp circuitry *****
Q1 1 in_m 12 qnpn
Q2 2 in_p 12 qnpn
Q3 12 5 3 qnpn
Q4 out 5 4 qnpn 10
Q5 5 5 5b qnpn 10
Q6 1 1 6 qnpn
Q7 2 1 7 qnpn
Q8 out 2 8 qnpn 20
R3 3 Vee 600
R4 4 Vee 60
R5a 0 5 4.24k
R5b 5b Vee 60
R6 Vcc 6 1k
R7 Vcc 7 1k
R8 Vcc 8 50
Cc out 2 1pF

*****
.ends
```


Simulation results for presented OA (in open loop without load)

| Parametr | Unit | Value for ideal OA | Value found | SPICE simulation command and PROBE post process view |
|----------------------------------------------------------------------------|----------|--------------------|-----------------------------------------------------------|----------------------------------------------------------------------------------------------|
| Differential voltage gain, A_D | V/V | ∞ | 9513 | .dc Vid -10m 10m 1u D(v(out)) for vid = vinoff |
| Common mode gain, A_{CM} | V/V | 0 | 0.331 | .dc Vcm -100m 100m 100u D(v(out)) for vid = vinff and Vcm = 0 |
| CMRR, $ A_D/A_{CM} $ | | ∞ | 28740 = 89,1dB | |
| PSRR ₊ , $ A_D/A_{VDD} $ PSRR ₋ , $ A_D/A_{VSS} $ | | ∞ | 9513/0.701 = 13571 = 82.7dB 9513/1.625 = 5854 = 75.3dB | .dc Vcc 4.5 5.5 10m or .dc Vee 4.5 5.5 10m D(v(out)) for vid = vinff and Vcc=5V or Vee=5V |
| Output voltage swing, V_{OMAX}, V_{OMIN} | V | ∞ | -4.97 – 4.90 | .dc Vid -100m 100m 1u v(out) |
| Input resistance, R_{ID} | Ω | ∞ | 58.06k | .dc Vid -10m 10m .01m -1/D(i(Vid)) dla Vid = vinff |
| Output resistance, R_{OUT} | Ω | 0 | 81.56k | .dc Iro -10u 10u .01u D(v(out)) dla vid = vinff i Iro = 0 |
| 3dB passband, f_{3dB} GBW | Hz | ∞ | 139.9 1.1M | .ac dec 100 1 1G Vdb(out) assuming dc Vid=vinoff |
| Phase margin | $^\circ$ | 180 | 44.3 | .ac dec 100 1 1G |
| Gain margin | dB | ∞ | 13.9 | Vdb(out) and vp(out) assuming dc Vid=vinoff |
| Slew rate | V/us | ∞ | 0.3593 -0.4829 | .tran 10n 500u 0 10n V(out) |
| Input referred offset voltage, V_{IOFF} | V | 0 | 8.278m | .dc Vid -10m 10m 1u v(out); vinoff = vid for 0 at output of OA |

Parameters to be found in voltage follower conf. with 5k load

- Voltage gain for low frequencies (A),
- Input resistance (R_{ID}),
- Output resistance (R_{OUT}),
- 3dB passband,
- Slew Rate factor (SR),
- THD for sine wave of amplitude 4V and frequency equal to 10kHz.

Circuit schematic



Circuit netlist / SPICE commands – tested circuit and testbench splited into separate files

Testbench file:

Operational Amplifier example

```
.lib op_amp_sub.lib
```

```
X1 in out out vcc vee op_amp
```

```
*** Power supply *****
```

```
Vcc Vcc 0 5V
```

```
Vee 0 Vee 5V
```

```
*****
```

```
*** Input source *****
```

```
Vi in 0 dc 0 ac 1 sin(0 4 10k)
```

```
+;pulse(-2 2 1u 1p 1p 50u 100u)
```

```
*****
```

```
*** Output resistance measurement ***
```

```
Iro 0 out DC 0
```

```
*****
```

continued ...

```
*** Load *****
```

```
Rload out 0 5k
```

```
*****
```

```
*** Library files (models) ***
```

```
.lib modele8.lib
```

```
*****
```

```
*** simulations commands ***
```

```
*.dc Iro -100u 100u .1u
```

```
*.dc vi -1 1 1m
```

```
*.ac dec 100 1 1G
```

```
*.tran 10n 500u 0 10n
```

```
.tran 1u 1m 0 1u
```

```
.four 10k 20 v([out])
```

```
*****
```

```
*** Options and output set ***
```

```
.probe
```

```
*****
```

```
.end
```

Opamp file:

```
*** Operational Amplifier subcircuit
```

```
.subckt op_amp in_p in_m out vcc vee
```

```
*** Op Amp circuitry *****
```

```
Q1 1 in_m 12 qnpn
```

```
Q2 2 in_p 12 qnpn
```

```
Q3 12 5 3 qnpn
```

```
Q4 out 5 4 qnpn 10
```

```
Q5 5 5 5b qnpn 10
```

```
Q6 1 1 6 qnpn
```

```
Q7 2 1 7 qnpn
```

```
Q8 out 2 8 qnpn 20
```

```
R3 3 Vee 600
```

```
R4 4 Vee 60
```

```
R5a 0 5 4.24k
```

```
R5b 5b Vee 60
```

```
R6 Vcc 6 1k
```

```
R7 Vcc 7 1k
```

```
R8 Vcc 8 50
```

```
Cc out 2 1pF
```

```
*****
```

```
.ends
```

Simulation results for presented OA (in follower configuration with 5k load)

| Parametr | Unit | Value for ideal OA | Value found | SPICE simulation command and PROBE post process view |
|-----------------------------------|----------|--------------------|-----------------|-----------------------------------------------------------------------------------------------------------|
| Voltage gain, A | V/V | 1 | 0.9982 | .dc Vi -5 5 1m D(v(out)) for vi = 0 |
| Voltage across input nodes of OA | V | 0 | 8.271m | .dc Vi -5 5 1m v(in,out) for vi = 0 |
| Input resistance, R_{ID} | Ω | ∞ | 19.97M | .dc Vi -1 1 1m -1/D(i(Vi)) dla Vi = 0 |
| Output resistance, R_{OUT} | Ω | 0 | 8.57 | .dc Iro -10u 10u .01u D(v(out)) dla vi = 0 i Iro = 0 Rload should be commented out |
| 3dB passband, f_{3dB} | Hz | ∞ | 1.877M | .ac dec 100 1 1G Vdb(out) assuming dc Vid=0 |
| Slew rate | V/us | ∞ | 0.343 -0.454 | Vi in 0 pulse(-2 2 1u 1p 1p 50u 100u) .tran 10n 500u 0 10n V(out) |
| THD @ 4V amp. and 10kHz sine wave | % | 0 | 0.3479 | Vi in 0 sin(0 4 10k) .tran 1u 1m 0 1u .four 10k 20 v([out]) View output file *.out for THD value |

Appendix – modele8.lib file

```
.model qnpn      NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=118.6 Ne=1.236
+               Ise=14.34f Ikf=.2524 Xtb=1.5 Br=7.134 Nc=2 Isc=0 Ikr=0 Rc=1
+               Cjc=9.393p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75
+               Tr=50.96n Tf=409.2p Itf=.6 Vtf=1.7 Xtf=3 Rb=10)
.model qpnp      PNP(Is=650.6E-18 Xti=3 Eg=1.11 Vaf=115.7 Bf=70.35 Ne=1.829
+               Ise=180.5f Ikf=1.079 Xtb=1.5 Br=4.146 Nc=2 Isc=0 Ikr=0 Rc=.715
+               Cjc=29.52p Mjc=.5383 Vjc=.75 Fc=.5 Cje=19.82p Mje=.3357 Vje=.75
+               Tr=119.9n Tf=757.7p Itf=.65 Vtf=5 Xtf=1.7 Rb=10)
```